
まえがき

この文書はMacaulay2の入門用に書いたものである. より詳細は Schenk
の本 [3]や Eisenbud等の本 [1]を参照されたい.
尚,Schenkの本はMacaulay2を用いて代数学を学ぶ為の本の為,数学のお
勉強に重点が置かれているのに対し,Eisenbud等の本はMacaulay2を使って
代数や代数幾何を料理する本の為,Macaulay2の言語や使い方に重点が置かれ
ている.
又,Macaulay2では Singularで用いられているライブラリが利用されてい
る. この Singularに関しては,Greuel-Pfisterの本 [2]を参照されると良い.こ
ちらは Singularを使った代数学の教科書である.

平成 18年 10月 25日 (水) 横田博史

第1章 Macaulay2の簡単な紹介

1.1 Macaulay2の概要

Macaulay2は可換環向けの数式処理システムである.Macaulay2は Singular
の幾つかのライブラリを用いてる為に Singularと似た点も多くある. その一
方で,Macaulay2の構文にはMathematica風なものも見受けられ,Singularと
は別の面白味のある数式処理システムである.

Macaulay2の大きな特徴として挙げられる事は,Macaulay2はオブジェクト
指向の数式処理システムである事だろう.その為,函数はオブジェクトに束縛
されたメソッドとなる,従って,基本的に何かの処理を行う為にはオブジェク
ト生成を最初に行わなければならない.又,函数も対象が違えば結果も当然異
なる.具体的に述べると, 多項式やイデアルといったものの計算を行う為,そ
の親にあたる環を先ず生成しなければならない.又,演算子^は左側の被演算

子が行列で右側の被演算子が数値の場合は行列の羃となる.ところが,右側の
被演算子が整数リストの場合, 行列の列の取り出しの演算子になる.

Macaulay2は固有のフロントエンドを持たない.個々の結果はその都度ファ
イルに残す事も可能であるが,Mathematica や Maple の様に, 終了時にセッ
ションをそのままファイルに簡単に残せないのが初心者にとっては難点であ

ろう.その為,フロントエンドとして利用可能な Emacsや TeXmacsと併用し
た方が良い.特に,TeXmacsを利用すれば, レゾリューション等の図式や数式
が美しくレンダリングされて表示されるので楽しい.

Macaulay2は詳細なオンラインマニュアルを有する.尚,0.9.8より前のMacaulay2
では, help函数でしか参照出来ず,xterm等の仮想端末を利用する場合でも,
Emacs風の行の編集と履歴の利用が出来ない.そして,help函数はテキスト
の内容を一気に表示するものである.その為,xtermの様な仮想端末では表示
内容がバッファに入り切らない事もあるので,Emacsや TeXmacs等のアプリ
ケーションをフロントエンドとして使わなければ尚更不便になる.
これが 0.9.8になると Emacs風の行編集と履歴の利用が可能で,オンライ
ンマニュアルも help函数以外に Texintoを用いる infoHelp函数やMozilla
等のブラウザを用いる viewHelp函数の二つの函数が用意されており,何かと
使い勝手が向上している.

Macaulay2の日本語の文献は殆んどないのが現状で,この小冊子を除くと
唯一, Macaulay2の作者の一人である Dan Graysonへのインタビューが”数

学のたのしみ No.11,多項式環の視点：グレブナー基底”(日本評論社,1999年)
の”グレブナエンジンのプログラマ達”に掲載されている.
英語の書籍としては Eisenbud等の本 [1]と Schenckの本 [3]の二つが出て
いる.
最初の Eisenbud 等の本には Macaulay2 の入門から応用迄が収録されて
いるが, この本はどちらかと言えば, こちらは代数学, 代数幾何に詳しい方
で,Macaulay2を使って何かをしたい為に,Macaulay2の使い方を調べるのに
適した本だろう.
次の Schenckの本の方は,Macaulay2を用いた代数幾何,代数的位相幾何学
の入門書で, これらの事項が手堅く纏められた良いテキストである.この本は
Macaulay2で一寸した計算をしながら,代数学のお勉強をしたい方に向いて
いる.
この冊子は手早くMacaulay2の使い方を解説する事を目的としている.Macaulay2
の詳細に関しては,上述の本やMacaulay2のオンラインマニュアルを参照さ
れたい.

1.2 起動方法と入力

Macaulay2の起動方法はパスが通っていれば,xterm等の仮想端末にM2と
入力すると Macaulay2 が立ち上がる.KNOPPIX/Math を使っていれば, メ
ニュバーの左から二番目にある

√
x メニューから Macaulay2を選択すれば

良い.
すると図 1.1の様にMacaulay2が立ち上がる.

図 1.1: kterm上で利用

但し,M2は前述の様に編集機能を持ったフロントエンドではないので,GNU
Emacsや TeXmacs等と云った外部のプログラムと併用する事を勧める.
特に,TeXmacsをフロントエンドとして使う場合,Macaulay2で処理した結
果が美しくレンダリングされて表示されるので目の保養にも良い.TeXmacs
を用いる場合には, TeXmacsの挿入 (insert)からセッション (session)を選ぶ
と TeXmacsで利用可能なアプリケーションの一覧が表示されるので,その中
からMacaulay2を選択すれば良い.
図 1.2 に TeXmacs をフロントエンドにした様子を示している. 図 1.1 の

kterm上のMacaulay2と同じ操作を行っているが, TeXmacs上の方がきれい
に表示されている事が分かる.

Macaulay2を起動すると入力を促すプロンプトが表示される.入力プロン
プトは i1 : の様に入力行である事を示す iと入力番号の組合せで, 出力の
プロンプトも同様に出力 o1 : の様に出力行である事を示す oと入力番号

の組合せとなっている.
尚,Macaulay2では入力行末尾にセミコロン";"を入れると入力に対するエ

コーバックが無効になる.それ以外では,入力に対して値やデータ型のエコー
バックを行う.

図 1.2: TeXmacs上で利用

Macaulay2は詳細なオンラインマニュアルを持っている.最初にオンライ
ンマニュアルを読む函数として help函数がある.help函数は help とだけ入
力すれば,help函数の概要が表示される.尚,help函数は一気に内容を表示する
為, xterm等の仮想端末でMacaulay2を起動した場合,ファイルの内容がバッ
ファに入り切らずに頭が読めなくなる事がある.その為,help函数しか使えな
い古いMacaulay2や,xterm等の仮想端末でどうしても使いたい場合,Emacs
上で利用する事を勧める.

Macaulay2-0.9.8では,普通のテキスト形式のマニュアルを表示する help函

数の他に,Texinfoを用いるinfoHelp函数,HTMLブラウザを利用するviewHelp

函数が使える.
ここでは Firefox等の HTMLブラウザを用いる viewHelpについて簡単に

述べよう. 尚,引数は help,infoHelp,viewHelpも全て同じである.単純にオン
ラインマニュアルが通常の Text,texinfoか HTMLかの違いが生じる程度で
あるが,今風に viewHelp を使うのが便利が良いだろう.
先ず, viewHelp とするとMacaulay2とそのパッケージに関するオンライン

マニュアルの見出しが表示される. viewHelp "help" で help函数のマニュア

ルを表示する.これでhelp函数の使い方が判るだろう. 基本的に viewHelp ”〈事項 〉”
で調べたい 〈事項 〉 のマニュアルが表示される.
最後にMacaulay2を終了する場合,quitと入力すれば良い.

1.3 基本的なデータ型について

Macaulay2では Singularと同様に最初に環を定義し,その環上で様々な対
象を構築して行く.Macaulay2はオブジェクト指向の処理言語の為,オブジェ
クトを生成しなければ, メソッドが使えない.これを具体的に言うと,最初に
環を定義しなければ,多項式や写像は定義出来ないし,これらを処理する函数
も使えない事になる.
尚,Symbol,Sequence,文字列や数値 (整数,有理数,実数),そして trueや false
で表現される真偽値と云った基本データ型,これらのデータで構成されたリス
ト,配列,行列の入力と処理は可能である.更に,それらを処理する函数も定義
可能である.

1.3.1 数

Macaulay2で扱える数値は,整数,有理数,実数である. Macaulay2では様々
な対象を定義する為に,予めその対象が含まれる環を定義する必要があるが,
整数環 Z,有理数環 Qと実数環 Rは最初から Macaulay2に組込まれた環と
なっている.これらの環はMacaulay2では各々ZZ,QQ, RRで表記される.
従って,これらの環に含まれる数値に関しては,そのまま入力が可能であり,
それらの数値に対しては四則演算も出来る.
数値の入力例を以下に示しておく:

i1 : 1

o1 = 1

i2 : 1+2

o2 = 3

i3 : 2*3

o3 = 6

i4 : 3/5

3

o4 = -

5

o4 : QQ

l5 : 2.9

o5 : 2.9

o5 : RR

この例で示す様に整数の計算の場合は結果だけが表示されているが,入力が
有理数と浮動点小数になれば,その計算で用いたデータ型 (この場合は属する
環)を計算結果に続けて表示する.

Macaulay2では四則演算は CやMaple等の数式処理と同様の演算子が利
用可能である.

四則演算¶ ³
演算子 例 概要

+ a + b aと bの和
- a - b aと bの差
* a * b aと bの積
/ a / b aの bによる商
^ a ^ b 羃乗 abµ ´

尚,FORTRANで用いられる羃乗の演算子**はMacaulay2では使えない.

i4 : 1+2*3-4^2/5

19

o4 = --

5

o4 : QQ

i5 : 1+0*1.9-2/5

o5 = 0.6

o5 : RR

この例の様に数値のみであれば電卓風に利用可能である.

その他の数値演算子として階乗!がある.ここで,演算子!を複数続ける事が

可能である.例えば, 4!!は (4!)!と解釈される.

1.3.2 数値函数

Macaulay2には cos等の三角函数や expや logといった指数函数や対数函

数が組込まれている.これらの函数はMaple等の函数とは異なり, 実数環 R
上の数値函数である.

i20 : cos(pi/2)

o20 = 6.12303*10^-17

o20 : RR

i21 : exp(1)

o21 = 2.71828

o21 : RR

i22 : log(1)

o22 = 0.

o22 : RR

これらの数値函数は一般の環上に使える様に拡張されていない為,xを変数
として含む多項式環を定義しても,textttsin(x)は xに数値が割当てられてい

ない為にエラーを返す.更に,多項式の微分を行う diff函数の引数として, 数
値函数を引渡しても,sinの変数が数値でない事に加え,sin(x) 自体が多項式
でない為にエラーになる.

1.3.3 論理式と論理演算

Macaulay2で真偽値は trueと falseで表現される.
ここでMacaulay2の論理式は,評価を行った結果として trueか falseを返
す式の事である.
代表的な論理式としては,a==1の様に比較の演算子を用いたものが挙げら
でる.以下にMacaulay2の持つ比較の演算子を示す.

比較の演算子¶ ³
演算子 構文 真となる条件

== a==b aと bが等しいければ真

>= a>=b aが b以上であれば真

> a>b aが bより大であれば真

<= a<=b aが b以下であれば真

< a<b aが bより小であれば真µ ´
i21 : 3==2

o21 = false

i22 : 3>=2

o22 = true

i23 : 3<4

o23 = true

尚,Macaulay2には面白い演算子?がある.この演算子は, a ? b とする事

で,aと bを比較した場合に真を返す比較の演算子を返す函数である.

i26 : 3 ? 3

o26 = ==

o26 : Keyword

i27 : 3 ? 5

o27 = <

o27 : Keyword

Macaulay2の論理演算子を以下に示しておく.

論理演算子¶ ³
演算子 構文 概要

and a and b 論理式 aと bの論理積
or a or b 論理式 aと bの論理和
not not(a) 論理式 aの否定

not a 論理式 aの否定 (小括弧を外した表記)µ ´
以下に具体例を示す.

i20 : 3>1 and 3<2

o20 = false

i21 : 3>1 or 3<2

o21 = true

i22 : not(3<1)

o22 = true

尚,notに関しては,not 3<1の様に小括弧 ()を外しても構わない.
ここで,論理演算子 andに関しては,論理式 a or bに含まれる論理式 aが

偽の場合,論理式 bの評価をMacaulay2は行わない. andと同様に論理演算子
orも,a or bを構成する論理式 aが真の場合,論理式 bの評価をMacaulay2
は行わない仕様となっている.

1.3.4 表象と文字列

数値の他に,Macaulay2では表象 (Symbol) と文字列 (String) が扱える.
先ず,表象はアルファベットで開始するアルファベットと数値と一部の記号
で構成される文字の羅列である.但し,Macaulay2の処理言語に含まれる for
等の文字の列は表象にならない.表象はMacaulay2で変数として利用可能で
ある.
次に,文字列は二重引用符"で括られた文字の列である.文字列の場合,"以外
に用いてはならない文字はなく,表象とは全くの別物である.尚,"を文字列中
に含める場合,\を用いる. 例えば, \” の様に表記すると良い.
以下に表象と文字列の例を示す.

i20 : x

o20 = x

o20 : Symbol

i21 : XYZ

o21 = XYZ

o21 : Symbol

i22 : w1 = "mike"

o22 = mike

i23 : w2 = "neko"

o23 = neko

この例で示す様に,変数に値を割当てる場合,演算子= を用いる.
文字列に対しては二つの演算子| と||がある.共に文字列の結合を行って新
しい文字列を生成する演算子であるが,改行コードが入るか入らないかの違い
がある.

以下に,!と||の演算子の違いを示しておく.

i14 : "今日はとても"|"良い天気"

o14 = 今日はとても良い天気

i15 : "今日はとても"||"良い天気"

o15 = 今日はとても

良い天気

i16 : "今日はとても"||""||"良い天気"

o16 = 今日はとても

良い天気

i17 : "\"三毛" | "猫\""

o17 = "三毛猫"

この例で示す様に,|の場合は単純に文字列を結合するだけであるが, ||の
場合は二つの文字列を結合するだけではなく,文字列の間に改行が入る.これ
らの演算子は結果の表示等で利用すると良いだろう.
又,前述の様に文字列中に二重引用符を入れたければ,二重引用符を \” で
置換えると良い

1.3.5 列,リスト,配列と行列の定義

Macaulay2には基本データ型で構成された列 (Sequence) , リスト (List),
配列 (Array) ,行列 (Matrix) がある. 先ず,列,リストと配列の定義方法を以
下に纏めておく.

列,リスト,配列¶ ³
列 (a1, · · · , an) 対象をコンマ,で区切って小括弧で括ったもの

リスト a1, · · · , an 列の中身を中括弧{}で括ったもの

配列 [a1, · · · , an] 列の中身を大括弧 []で括ったものµ ´
次に列,リスト,配列と行列の例を示す:

i27 : (1,2,3)

o27 = (1, 2, 3)

o27 : Sequence

i28 : 1..5

o28 = (1, 2, 3, 4, 5)

o28 : Sequence

i29 : (1,2.4,5/7)

5

o29 = (1, 2.4, -)

7

o29 : Sequence

i30 : {1,2,3,54}

o30 = {1, 2, 3, 54}

o30 : List

i31 : [1,2,3,4,5]

o31 = [1, 2, 3, 4, 5]

o31 : Array

Macaulay2の面白い機能に,1づつ増加する数列を簡単に生成する事が容易に
行える事がある,この場合,演算子..を用いる. 演算子..は 〈開始値 〉..〈終値 〉
で, 〈開始値 〉から開始して 1刻みで 〈終値 〉までの数値列を生成する.例え
ば,(1..5)で (1,2,3,4,5)が生成される. この演算子..は下付きの添字を表

現する演算子 と組合せる事で,x 1,x 2,..,x 10の様に添字付けられた表象列を

x 1..x 10と入力して容易に生成する事が可能になる.
以下に,演算子..の例を示す.

i1 : 1..4

o1 = (1, 2, 3, 4)

o1 : Sequence

i2 : 12..19

o2 = (12, 13, 14, 15, 16, 17, 18, 19)

o2 : Sequence

i3 : x_1..x_5

o3 = (x , x , x , x , x )

1 2 3 4 5

o3 : Sequence

列,リスト,及び配列の成分や長さを求める演算子に# がある.#の構文を以
下に示しておく.

演算子#¶ ³
構文 概要

〈a〉 # 〈整数 〉 〈a〉 の第 〈整数 〉 + 1番目の成分
# 〈a〉 〈a〉 の長さµ ´

列,リスト,配列の成分は Cと同様に 0から開始するので,配列 aの左から
第 5番目の成分を取出したい場合, a#4 と入力する事に注意されたい.
以下に演算子#の実行例を示しておく.

i50 : a=(1..10)

o50 = (1, 2, 3, 4, 5, 6, 7, 8, 9, 10)

o50 : Sequence

i51 : a#9

o51 = 10

i52 : b=[10,11,12,13,14,15]

o52 = [10, 11, 12, 13, 14, 15]

o52 : Array

i53 : b#2

o53 = 12

i54 : c={1,2,3,5}

o54 = {1, 2, 3, 5}

o54 : List

i55 : c#2

o55 = 3

i56 : #a

o56 = 10

i57 : #b

o57 = 6

i58 : #c

o58 = 4

列,リストと配列と異なり,行列は函数 matrix を用いて定義する.
以下にmatrix函数の例を示しておく.

i32 : A=matrix {{1,2,3,4},{5,4,3,2}}

o32 = | 1 2 3 4 |

| 5 4 3 2 |

2 4

o32 : Matrix ZZ <--- ZZ

i33 : B=matrix( {{1,1},{-1,2}} )

o32 = | 1 1 |

| -1 2 |

2 2

o33 : Matrix ZZ <--- ZZ

ここでの例では matrix函数を使って Aと Bの二つの行列を生成している.
これらの例で示す様に,Macaulay2の行列は線形写像として捉えられている.
この際に,行列 Aの生成では A=matrix 1,2,3,4,5,4,3,2で行列 Bの生
成では B=matrix( 1,1,-1,2 )としているのに対し, 行列 Aでは小括弧 ()

を外している事に注意されたい.この様にMacaulay2 の特徴として,一変数
函数の場合に限って,函数の括弧を省略する事が許容されている.
行列の演算子として|と||の二つがある.これらの演算子は文字列の演算子
としても現われているが,行列の演算子としても少し似た動作となるが別の働
きをする.
先ず,演算子|は与えられた二つの行列を水平方向に繋げた行列を生成し, 演
算子||は与えられた二つの行列を上下に結合する.

o66 = | 1 2 |

| 3 2 |

2 2

o66 : Matrix ZZ <--- ZZ

i67 : x2=matrix{{0,1},{1,0}}

o67 = | 0 1 |

| 1 0 |

2 2

o67 : Matrix ZZ <--- ZZ

i68 : x1|x2

o68 = | 1 2 0 1 |

| 3 2 1 0 |

2 4

o68 : Matrix ZZ <--- ZZ

i69 : x1||x2

o69 = | 1 2 |

| 3 2 |

| 0 1 |

| 1 0 |

4 2

o69 : Matrix ZZ <--- ZZ

次に,行列から列や行を取出す場合,演算子 とˆを用いる.

演算子 とˆ¶ ³
〈行列 〉 {〈整数1〉, · · · , 〈整数n〉}
〈行列 〉 ˆ {〈整数1〉, · · · , 〈整数n〉}µ ´
行列の列や行を取出す演算子は演算子の左側に行列を置き,右側に取出す成
分のリストを配置する.ここで,行列に関しても成分番号は 1ではなく 0から
開始する事に注意されたい.

i125 : A1=matrix {{1,2,3},{0,1,3},{4,1,9}}

o125 = | 1 2 3 |

| 0 1 3 |

| 4 1 9 |

3 3

o125 : Matrix ZZ <--- ZZ

i126 : A1_{0}

o126 = | 1 |

| 0 |

| 4 |

3 1

o126 : Matrix ZZ <--- ZZ

i127 : A1_{0,2}

o127 = | 1 3 |

| 0 3 |

| 4 9 |

3 2

o127 : Matrix ZZ <--- ZZ

i128 : A1^{0}

o128 = | 1 2 3 |

1 3

o128 : Matrix ZZ <--- ZZ

i129 : A1^{0,2}

o129 = | 1 2 3 |

| 4 1 9 |

2 3

o130 : Matrix ZZ <--- ZZ

尚,演算子^は羃演算子としての特性を持つ為,演算子^の左側がリストでな

く数値や変数の場合は羃演算子として処理される事に注意されたい.
尚,演算子 に関してはリストの代りに数値を設定しても挙動に違いはない.

o137 = | 1 2 0 |

| 7 1 3 |

| 0 1 9 |

3 3

o137 : Matrix ZZ <--- ZZ

i138 : A1^{2}

o138 = | 0 1 9 |

1 3

o138 : Matrix ZZ <--- ZZ

i139 : A1^2

o139 = | 15 4 6 |

| 14 18 30 |

| 7 10 84 |

3 3

o139 : Matrix ZZ <--- ZZ

i140 : A1_{2}

o140 = | 0 |

| 3 |

| 9 |

3 1

o140 : Matrix ZZ <--- ZZ

i141 : A1_2

o141 = | 0 |

| 3 |

| 9 |

3

o141 : ZZ

1.3.6 リストと行列の演算

Macaulay2の面白い機能で,リストに対してはMATLABの様な四則演算
も出来なくはない点がある.演算は同じ長さで同じ書式のリスト同士で可能で
ある.

i26 : a={{1,2,3,4},{5,6}}

o26 = {{1, 2, 3, 4}, {5, 6}}

o26 : List

i27 : b={{3,1,2,4},{10,9}}

o27 = {{3, 1, 2, 4}, {10, 9}}

o27 : List

i28 : 2*a+b/5

13 21 32 44 69

o28 = {{--, --, --, --}, {12, --}}

5 5 5 5 5

o28 : List

この例ではリスト aと bを定義した後に,2*a+b/5 でリストの各成分に対
して 2 · a + b

5 の処理を行っている.
但し,MATLABやOctaveと比べると,Macaulay2はリストの処理に強いと
は言えない.
その一例として,積の順番の問題を示しておく.

i30 : b*2

stdio:30:2: expected pair to have a method for ’*’

i31 : 2*b

o31 = {{6, 2, 4, 8}, {20, 18}}

上の b*2と 2*bでは前者は失敗するものの,後者では正常に終了している.
この様にリストに対する積や羃乗に関しては注意が必要である.
この様に,MATLABの様な処理が行えるとは限らない.尤も,その様な処理
をMacaulay2 で行う事は考え難いが….
尚,リストに対しては他の数式処理と同様に,apply函数を用いて,Macaulay2
の函数を作用させる事が可能である.

i43 : apply({1,2,3,4,5},i->i^2)

o43 = {1, 4, 9, 16, 25}

o43 : List

i44 : apply({1,2,3,4,5},i->i^2*sin(i))

o44 = {0.841471, 3.63719, 1.27008, -12.1088, -23.9731}

o44 : List

ここで apply函数でリストに作用させる函数をMapleの様に演算子-> で

定義している事に注意されたい. この函数の定義方法に関しては 1.5小節に
て詳細を述べる.
次に,行列に対しては和,差,積として羃演算が可能である.

行列の演算子¶ ³
+ A+B 行列の和

- A-B 行列の差

* A-B 行列の積

ˆ A^n 行列の羃 (n ≥ 0)µ ´
但し,羃^に関しては右側の被演算子の挙動によって意味が異なる事に注意

されたい.

1.3.7 環

Macaulay2では環の定義は以下の構文で環を定義する.

環の定義¶ ³
〈環の名前 〉 = 〈係数環 〉[変数1, · · · ,変数n]µ ´
ここで係数環としては,デフォルトでMacaulay2に定義されている 3個の
環, ZZ(整数環 Z),QQ(有理数体 Q ),RR(実数体 R), 更に,利用者が定義した
環が利用可能である.
以下に具体的な環の定義例を示そう:

i1 : R=QQ[x,y,z]

o1 = R

o1 : PolynomialRing

i2 : R’=ZZ[x,y,z]

o2 = R’

o2 : PolynomialRing

i3 : poly1=x^2+2*y^2+3*z^2-1

2 2 2

o3 = x + 2y + 3z - 1

o3 : R’

この例では環Rの係数環に有理数体 Q(Macaulay2では QQと表記) を指定
し,変数を x, y, z の 3個を指定している.この指定を行うと,逆辞書式順序で
環が生成され,以下の処理では新たに加群や環を定義するまでは,ここで定義
した環 R 上で処理を行う.環の定義では変数の順序の指定等も行える. 勿論,
ZZ7,即ち,ZZ/7の様な環も定義可能である.
尚,係数環によっては使えない函数が存在するので注意が必要である.例え
ば,与えられたイデアルのGroebner基底を求める際に gb函数を用いるが,こ
の gbは整数環 Z では使えない (係数の割算が出来ない為!).
環を定義すると,自動的にその生成した環にポインタが移動する.即ち,新
しく生成した環で多項式等の対象を生成し,操作する事となる.

ここで,複数の環がある環境で,既存の別の環に環を切り替える場合, use函
数を用いる.因に Singularでは setring命令を用いる.

i5 : R=ZZ[x,y,z]

o5 = R

o5 : PolynomialRing

i6 : R2=QQ[x,y]

o6 = R2

o6 : PolynomialRing

i7 : use R

o7 = R

o7 : PolynomialRing

但し,Singularではポインタが置かれた環 (基礎環と呼ぶ)上で様々な対象
を生成するが,Macaulay2も同様である.但し,Singularと違い,Macaulay2で
は環毎に同じ変数名を持つ対象を持たせる事は出来ない.

i1 : R1=QQ[x,y,z];

i2 : R2=QQ[x,y,a];

i3 : R3=QQ[x,a,b];

i4 : use R1;

i5 : f=x+y+z;

i6 : use R2;

i7 : f=x*y*a^2;

i8 : use R3;

i9 : f=x*a^2+b;

i10 : use R1;

i11 : f

2

o11 = x*a + b

o11 : R3

この例では,環 R1,R2と R3を定義し,各々に多項式を変数 fに割当ててい
る. 結局,変数 fの内容は環毎に保存されない為,変数 fの内容の書換が行わ
れている事が判る.この様にMacaulay2では一つの変数に複数の意味を持た
す事が出来ないが, その一方で, 基礎環が異っていても対象の内容をエコー
バックで確認する事が可能となる.この様子は,Singularが垣根で小さく区切
られた庭園の様になっており, 同じ名前の物は各区画に一つ置く事が出来る
が,Macaulay2の場合は広々とした公園なので,同名の物を置く事が許容され
ないとでも例えられるだろうか.
ある対象がどの環に属するものかを調べたければ,対象の名前を入力して,
そのエコーバックで判断する事も可能であるが,それとは別に,ring函数を用
いる事でも出来る.この函数はその対象が定義された環の名前を返す函数で
ある.
ある環で構築した対象を別の環に代入する函数が substitute である.こ
の函数 substitute の引数は対象と複写先の環を指定する.

i1 : R0=ZZ[x,y,z,w]

o1 = R0

o1 : PolynomialRing

i2 : poly1=x^2+2*y^2+3*z^2-1

2 2 2

o2 = x + 2y + 3z - 1

o2 : R0

i3 : R=ZZ[x,y,z]

o3 = R

o3 : PolynomialRing

i4 : poly2=substitute(poly1,R)

2 2 2

o4 = x + 2y + 3z - 1

o4 : R

i5 : poly1

2 2 2

o5 = x + 2y + 3z - 1

o5 : R0

i6 : poly2

2 2 2

o6 = x + 2y + 3z - 1

o6 : R

この例では最初に環 R0= Z[x, y, z, w] を生成し, 環 R0 上の多項式 poly1
を定義している. それから環 R=Z[x, y, z] とその環 R 上の多項式 poly2 を
substitute函数を用いて環 R0の多項式 poly1で定義している.
尚,定義した対象の詳細は describe函数 で調べる事が可能である.

i1 : R=ZZ/5[x,y]

o1 = R

o1 : PolynomialRing

i2 : describe R

25



ZZ

o2 = -- [x, y]

5

i3 : f=(x+2*y+3)^5

5 5

o3 = x + 2y - 2

o3 : R

i4 : g=(x,y)->x*y+2

o4 = g

o4 : Function

i5 : describe R

ZZ

o5 = -- [x, y]

5

i6 : describe f

5 5

o6 = x + 2y - 2

この例では describe函数を用いて環や多項式の情報を表示させている.

1.4 文

Macaulay2の文は,a=1の様に値を割当てる文,単純に 1+2*3や 3<1の様に

Macaulay2を使って解釈を行うものであ!る.
Macaulay2では文を小括弧 ()を用いてグループ化する事も可能である. 文
のグループ化は制御文や函数の定義で用いられる.

1.4.1 制御文

Macaulay2の制御文には,if文, for文や while文がある.

制御文¶ ³
制御文 構文

if if 〈条件文 〉 then 〈処理1〉
if 〈条件文 〉 then 〈処理1〉 else 〈処理2〉

for for 〈制御変数 〉 = 〈初期値 〉 to 〈終値 〉 do 〈処理 〉
for 〈制御変数 〉 = 〈初期値 〉 to 〈終値 〉 list 〈処理 〉

while while 〈条件文 〉 do 〈処理 〉
while 〈条件文 〉 list 〈処理 〉µ ´

ここでの条件文は解釈される事で trueか falseが返却されるMacaulay2の
文である. 〈処理 〉 はMacaulay2の文,複数の文を小括弧 ()を使ってグルー

プ化した文である.
先ず,Macaulay2の if文は Cの if文と同じ構文である. 以下に簡単な例を
示しておこう.

i14 : i=random(10);if even i then a1=1 else a1=2

o15 = 2

Macaulay2の for文は,doか listでその挙動が異なる.先ず,doの場合は処
理を実行させ, listの場合には処理した結果をリストにして返却する.

i61 : for i from 1 to 5 do print(i*5)

5

10

15

20

25

i62 : for i from 1 to 5 list i*5

o62 = {5, 10, 15, 20, 25}

o62 : List

これは whileも同様である.

i69 : i=1;while i<5 do (i=i+1;print i^5)

32

243

1024

3125

i71 : i=1;while i<5 list (i=i+1; i^5)

o72 = {32, 243, 1024, 3125}

o72 : List

whiteや forループを抜ける函数として breakがある. breakは単体,或い
は一つの引数を持つ.単体の場合,breakは何も返さないが, 引数を指定した場
合,breakはその引数の値を返す.

i18 : for i from 2 to 100 do if not isPrime i then break i

o18 = 4

i19 : i=1;while i<=100 do (i=i+1;if not isPrime i then break)

i20 :

どちらも同じ処理を行っているが,while文の場合は breakの引数が無い為
に, 引数 (この場合は iの値)が返されていない.

1.5 函数

Macaulay2には函数 (Function)がある.Macaulay2の函数定義では以下に
示す様に演算子->を用いる.

函数定義の構文¶ ³
〈函数名 〉 = (〈変数1〉, · · · , 〈変数n〉) -> (〈文1〉; · · · 〈文n〉; )µ ´
ここで文は a=1の様な変数の割当や制御文等であり,複数の文はセミコロ
ン;で区切ったものになる.全体的な雰囲気はMapleの函数定義に似ている.
以下に具体的な例を示しておこう.

i32 : f=(x,y,z) -> x^2+y/z

o32 = f

o32 : Function

i33 : f(1,2,3)

5

o33 = -

3

o33 : QQ

この例では単純に (x, y, z) → x2 + y/z の変換を行う.Macaulay2には演算
子:=があり,この記号を使って函数内部で変数を割当てると,その変数は函数
内部の局所変数として解釈される.

i4 : f=(x,y,z)->a:=x+y^3;

i5 : f(1,2,3)

o5 = 9

i6 : a

o6 = a

o6 : Symbol

i7 : g=(x,y,z)->a=x+y^3;

i8 : g(1,2,3)

o8 = 9

i9 : a

o9 = 9

最初の函数 fでは演算子:=を用いている為,aは函数内部の値が設定されず
に表象のままである. 次の gの例では演算子=を用いている為, aに値が設定
されている事に注意されたい.

1.6 M2ファイル

Macaulay2で利用可能な函数は m2ディレクトリに末尾が m2のテキスト
ファイルに記述されている.Macaulay2 ではこれらの m2 ファイルの読込に
load函数と needs函数を用いる.

m2ファイルはMacaulay2の複数の函数や変数を定義する事が可能である.
記述は基本的にMacaulay2に入力する書式で構わない.ここで,m2ファイル
内部での注釈行は演算子--0 えんさんし@演算子!--を先頭に置いた文字の

羅列である.この演算子--は,Macaulay2に直接入力する場合,演算子の前に
Macaulay2の通常の入力行を必要とするが,m2ファイルで用いる場合,その
様な制約はない.

m2ファイルの読込に用いる load函数と needs函数の構文を以下に示して

おく.
load函数と needs函数¶ ³

load ”〈ファイル名 〉”
needs ”〈ファイル名 〉”µ ´

load函数と needs函数は共に指定した m2ファイルをMacaulay2に読込
む函数であるが,load函数が単純に指定したファイルをMacaulay2に読込む
函数であるのに対し,needs函数は指定したm2ファイルが既に読込まれてい
る場合には,m2ファイルの読込を行わない函数である.
これらの函数は大域変数 path に設定されたリストに含まれるディレクト

リから指定されたファイルを検索し, このリストに含まれていればファイルの

30



読込を行う.従って,大域変数 path に含まれるディレクトリ上にファイルが

存在する場合,これらの函数の引数としては, ファイル名のみで構わない.し
かし,ファイルが大域変数 pathに含まれない場合には,フルパスでファイル
を指定しなければならない.
code函数はMacaulay2の函数がどのm2ファイルに収録されたもので,具
体的な函数の定義内容を表示する函数である.この code函数はm2ファイル
の中身を表示する為,m2ファイルを記述せずに定義した函数に対しては効力
が無い.

1.7 ファイル出力

Macaulay2からファイルへの書込みでは,演算子<< を用いる.
この演算子<<は通常出力とファイルの出力の指定が行える.左側被演算子
を持たない場合は,右側の被演算子をMaculay2で解釈した結果を標準出力に
出力する. 左側に文字列を置いた場合,出力先は文字列で指定されたファイル
となり. <<の右側被演算子に closeが指定される迄,同じファイルのままとな
る. 尚,<<は原始的な出力函数の為,単体で用いられた場合,ファイルの先頭か
ら書き込みを行う.
更に,この演算子<<による出力の影響は普通の函数には及ばない.例えば,

print函数で表示した結果は<<によってファイルに切り替えられる訳ではない.
その為,綺麗な書式のファイルを生成する事は出来ない.一時的に結果を残
す程度の事しか出来ない.

第2章 特異点遊戯

2.1 はじめに

この章ではMacaulay2を使った具体例として,曲線や曲面の特異点を計算
させてみよう.

2.2 Macaulay2による特異点の計算

曲線や曲面上のある点が,その点での偏微分が全て 0 になる場合,この点を
特異点と呼ぶ.この章ではMacaulay2を使って特異点を探して遊ぶ事にする.
最初に,平面曲線 x3−y2 = 0を考えよう.この曲線の X,Y 方向の偏微分は

各々, 3x2 と −2y である.従って,この平面曲線の特異点は原点 (0, 0) になる.
次の図 2.1に surfで用いて描いたこの曲線のグラフを示す.

図 2.1: x3 − y2 = 0のグラフ

上のグラフから原点付近が縊れている事が判る.
この特異点の計算をMacaulay2で行う.先ず,曲線を多項式 f の零点集合

として与え,多項式とその偏微分から生成されるイデアル I を定義する. す
ると,この多項式とその偏微分から生成されるイデアルの零点集合 V (I) が多
項式 f で与えられる曲線の特異点集合となる.
ここで多項式環の係数体が代数的に閉じている場合, V (I) を零点集合とす
るイデアルは I の根基イデアル (radical)

√
I に一致する事が知られている

(Hilbeltの零点定理).従って,特異点を調べたければ先ず根基イデアル
√

I を

計算し,その性質を調べれば良い事になる.
ここで環 R のイデアル I の根基

√
I は,環 R のイデアルで, g ∈

√
I であ

れば, ある自然数 n(≥ 1) に対して, gn ∈ I となる性質を持つものである.例
えば,イデアル I が多項式 x2 + 2x + 1 で生成されている場合, I の根基

√
I

は x + 1 で生成されるイデアルとなる. イデアル I の根基
√

I は I ⊂
√

I を

満すイデアルでもある.この根基の別の表現は,イデアル I を包含する全ての

素イデアルの共通部分集合として表現される.
では,実際に x3 − y2 = 0 をMacaulay2を使って遊んでみよう.

i1 : R=QQ[x,y];

i2 : f=x^3-y^2;

i3 : neko=ideal {f, diff(x,f), diff(y,f)}

3 2 2

o3 = ideal (x - y , 3x , -2y)

o3 : Ideal of R

i4 : radical neko

o4 = ideal (y, x)

この例では,最初に有理数体 Q を係数環に持つ環 R とその環上の多項式

f を定義している.この様にMacaulay2では多項式の処理を行う前に必ず親
の環を定義しなければならない.その為,環を定義して初めて多項式や多項式
で生成するイデアル等が扱える様になる.
ここでは多項式 f とその偏微分から生成されるイデアル neko を定義する.
多項式は多くの数式処理と同様の表記で入力すれば良い.但し,変数への代入
は大域変数として利用するのであれば演算子=を用い,Macaulay2のプログラ
ムで局所的に利用するのであれば,演算子:=を用いる.
イデアルの定義では函数 idealを用いる. この ideal函数の構文を以下に

示しておく.
ideal函数の構文¶ ³

ideal (〈多項式1〉, 〈多項式n〉)
ideal(〈多項式リスト 〉)
ideal(〈多項式行列 〉)µ ´

基礎環上の多項式,多項式の列にリスト,多項式行列等からイデアルは定義
可能である.ここで,函数 idealの小括弧 ()は最初の構文の様に複数の多項

式をコンマで区切る場合を除いて省略可能である.
そして,多項式の微分では diff函数を用いる.この diff函数の構文も以下

に示しておこう.
diff函数の構文¶ ³

diff(〈変数 〉, 〈多項式 〉)
diff(〈変数 〉, 〈行列 〉)µ ´
この様に,微分の函数 diffはMaple,Mathematica,Maximaとは違い, 最初
の引数が変数,その次に多項式となる事に注意されたい. 更に,diff函数で微
分可能なものは多項式と多項式を成分とする行列に限定される.即ち,cosの
様なMacaulay2組込の数値函数の微分は出来ない.
イデアルの根基は radical函数で計算出来る.この radical函数は以下に

示す様に引数にイデアルを取り,. 根基を返すが,この根基はイデアルである.
radical¶ ³

radical(〈イデアル 〉)µ ´
この例では,x3 − y2,3x2,−2yで生成されるイデアル nekoの根基が (y, x) で
生成されるイデアルとなる事が分る.この根基の零点集合は x = 0, y = 0 の
みである.以上から, (0, 0) がこの曲線の特異点となる事が分る.
今度は任意の有理数 t に対して,x3 − y2 + t を考えよう.ここでも上と同様
の方法で根基イデアルを計算しよう.

i6 : R=QQ[x,y,t]

o6 = R

o6 : PolynomialRing

i7 : f=x^3-y^2+t;

i8 : neko = ideal { f, diff(x,f), diff(y,f) }

3 2 2

o8 = ideal (x - y + t, 3x , -2y)

o8 : Ideal of R

i9 : radical neko

o9 = ideal (t, y, x)

o9 : Ideal of R

上記の結果から, t が零でなければ x3 − y2 + t には特異点が無い事が分る.
実際,図 2.2は surfを用いて t = 1 の場合を可視化したものであるが, この
様に滑らかな曲線となっている事が判る.

図 2.2: t = 1 の場合

索 引

演算子

*, 7
+, 7
-, 7
.., 14
/, 7
=, 11
#, 14
ˆ, 7
->, 21
:=, 29
<<, 31
and, 10
not, 10
or, 10
！, 8

函数

I
ideal, 33

記号

QQ, 6, 22
RR, 6, 22
ZZ, 6, 22

制御文

for, 27
if, 27
while, 27

大域変数

P
path, 30

データ型

false, 9
true, 9

函数, 29
行列 (Matrix), 13
シンボル, 11
配列 (Array), 13
表象, 11
文字列, 11
リスト (List), 13
列 (Sequence), 13

函数

A
apply, 21

C
code, 31

D
describe, 25

L
load, 30

M
matrix, 16

N
needs, 30

R
radical, 34
ring, 24

S
substitute, 24

U
use, 23

き

局所変数, 29
し

実数, 6
せ

整数, 6
ほ

ポインタ, 22
ゆ

有理数, 6
り

リストの処理, 20

